Section 6.4 examples using Maple.

New commands:
6.1: Maple can perform a Laplace transform and inverse transform: laplace(f(t),t,s)
invlaplace(F(s),s,t)
We must call in "inttrans" (for integral transforms): with(inttrans)

6.3, 6.4: u(t-c) from the text is Heaviside(t-c) in Maple.

Exercises from 6.3 and 6.4:

```
> with(plots):
  with(inttrans):
```

6.3.15: Find the Laplace transform of the function f(t) (given in piecewise form)

```
> f:=(Heaviside(t-Pi)-Heaviside(t-2*Pi))*(t-Pi);
```

$$f := (\mathrm{Heaviside}(t-\pi) - \mathrm{Heaviside}(t-2\pi))\,(t-\pi) \quad (1)$$

```
> plot(f,t=0..8);
```




```
> F:=laplace(f,t,s);
```

$$F := \frac{e^{-s\pi} - (\pi s + 1)\, e^{-2 s\pi}}{s^2} \quad \textbf{(2)}$$

6.3.23: Find the inverse Laplace transform of the given expression

```
> F:=(s-2)/(s^2-4*s+3);
```

$$F := \frac{s-2}{s^2 - 4s + 3} \quad \textbf{(3)}$$

```
> F1:=convert(F,parfrac,s);
```

$$F1 := \frac{1}{2(s-3)} + \frac{1}{2(s-1)} \quad \textbf{(4)}$$

```
> F2:=invlaplace(exp(-s)*F1,s,t);
```

$$F2 := \frac{1}{2}\,\mathrm{Heaviside}(t-1)\,\left(e^{3t-3} + e^{t-1}\right) \quad \textbf{(5)}$$

Section 6.4 Examples follow (Exercises 3, 10, 17)

```
> #Exercise 6.4.3
  G:=sin(t)-sin(t-2*Pi)*Heaviside(t-2*Pi); #Forcing function
```

$$G := \sin(t) - \sin(t)\,\mathrm{Heaviside}(t - 2\pi) \quad \textbf{(6)}$$

```
> DE03:=diff(y(t),t$2)+4*y(t);
```

$$DE03 := \frac{d^2}{dt^2}\,y(t) + 4\,y(t) \quad \textbf{(7)}$$

```
> #We can solve the DE "manually" if we want-  Here is how you do
  it!
  Eqn03:=DE03=G;
```

$$Eqn03 := \frac{d^2}{dt^2}\,y(t) + 4\,y(t) = \sin(t) - \sin(t)\,\mathrm{Heaviside}(t - 2\pi) \quad \textbf{(8)}$$

```
> Eqn03A:=laplace(Eqn03,t,s);
```

$$Eqn03A := s^2\,laplace(y(t), t, s) - \mathrm{D}(y)(0) - s\,y(0) + 4\,laplace(y(t), t, s) = \frac{-e^{-2 s\pi} + 1}{s^2 + 1} \quad \textbf{(9)}$$

```
> Eqn03B:=solve(Eqn03A,laplace(y(t),t,s));
```

$$Eqn03B := -\frac{-y(0)\,s^3 - \mathrm{D}(y)(0)\,s^2 - s\,y(0) + e^{-2 s\pi} - \mathrm{D}(y)(0) - 1}{(s^2+1)\,(s^2+4)} \quad \textbf{(10)}$$

```
> Eqn03C:=subs({y(0)=0,D(y)(0)=0},Eqn03B);
```

$$Eqn03C := -\frac{-1 + e^{-2 s\pi}}{(s^2+1)\,(s^2+4)} \quad \textbf{(11)}$$

```
> Eqn03D:=invlaplace(Eqn03C,s,t);
```

$$Eqn03D := \frac{1}{6}\,\mathrm{Heaviside}(-t + 2\pi)\,(-\sin(2t) + 2\sin(t)) \quad \textbf{(12)}$$

```
> plot({G,Eqn03D},t=0..3*Pi);
```



Here we show the solution to Exercise 10 using Maple's built-in solver:

```
> G:=sin(t)*(1-Heaviside(t-Pi));
```

$$G := \sin(t)\,\left(1 - \mathrm{Heaviside}(t - \pi)\right) \tag{13}$$

```
> DE10:=diff(y(t),t$2)+diff(y(t),t)+(5/4)*y(t);
```

$$DE10 := \frac{d^2}{dt^2}\,y(t) + \frac{d}{dt}\,y(t) + \frac{5}{4}\,y(t) \tag{14}$$

```
> Y1:=simplify(rhs(dsolve({DE10=G,y(0)=0,D(y)(0)=0},y(t),method=
  'laplace')));
```

$$Y1 := \frac{32}{17}\,e^{-\frac{1}{4}t + \frac{1}{4}\pi}\,\mathrm{Heaviside}(t-\pi)\,\cos(t)\,\sinh\left(\frac{1}{4}\,t - \frac{1}{4}\,\pi\right) - \frac{8}{17}\,e^{-\frac{1}{4}t + \frac{1}{4}\pi}\,\mathrm{Heaviside}(t-\pi)\,\sin(t)\,\cosh\left(\frac{1}{4}\,t - \frac{1}{4}\,\pi\right) - \frac{32}{17}\,\cos(t)\,e^{-\frac{1}{4}t}\,\sinh\left(\frac{1}{4}\,t\right) + \frac{8}{17}\,\sin(t)\,e^{-\frac{1}{4}t}\,\cosh\left(\frac{1}{4}\,t\right) \tag{15}$$

```
> plot({G,Y1},t=0..3*Pi);
```



We might do this "manually" to see if Y1 is actually a nicer expression:

```
> Eqn10A:=DE10=G;
```

$$Eqn10A := \frac{d^2}{dt^2}\, y(t) + \frac{d}{dt}\, y(t) + \frac{5}{4}\, y(t) = \sin(t)\, \left(1 - \mathrm{Heaviside}(t - \pi)\right) \quad (16)$$

```
> Eqn10B:=laplace(Eqn10A,t,s);
```

$$Eqn10B := s^2\, laplace(y(t), t, s) - D(y)(0) - s\, y(0) + s\, laplace(y(t), t, s) - y(0) + \frac{5}{4}\, laplace(y(t), t, s) = \frac{e^{-s\pi} + 1}{s^2 + 1} \quad (17)$$

```
> Eqn10C:=subs({y(0)=0,D(y)(0)=0},Eqn10B);
```

$$Eqn10C := s^2\, laplace(y(t), t, s) + s\, laplace(y(t), t, s) + \frac{5}{4}\, laplace(y(t), t, s) = \frac{e^{-s\pi} + 1}{s^2 + 1} \quad (18)$$

```
> Eqn10D:=solve(Eqn10C,laplace(y(t),t,s));
```

$$Eqn10D := \frac{4\left(e^{-s\pi} + 1\right)}{\left(s^2 + 1\right)\left(4 s^2 + 4 s + 5\right)} \quad (19)$$

```
> F:=4/((s^2+1)*(4*s^2+4*s+5));
```

$$F := \frac{4}{(s^2+1)\,(4\,s^2+4\,s+5)} \qquad (20)$$

```
> F1:=convert(F,parfrac,s);
```

$$F1 := \frac{1}{17}\,\frac{64\,s+48}{4\,s^2+4\,s+5} + \frac{1}{17}\,\frac{-16\,s+4}{s^2+1} \qquad (21)$$

```
> Y1A:=invlaplace(F1,s,t);
```

$$Y1A := -\frac{16}{17}\cos(t) + \frac{4}{17}\sin(t) + \frac{4}{17}\,e^{-\frac{1}{2}t}\,(4\cos(t)+\sin(t)) \qquad (22)$$

```
> Y2A:=invlaplace(F1*exp(-s*Pi),s,t);
```

$$Y2A := \frac{4}{17}\,\mathrm{Heaviside}(t-\pi)\left(4\cos(t)-\sin(t)-e^{-\frac{1}{2}t+\frac{1}{2}\pi}\,(4\cos(t)+\sin(t))\right) \qquad (23)$$

```
> Y:=Y1A+Y2A;
```

$$Y := -\frac{16}{17}\cos(t) + \frac{4}{17}\sin(t) + \frac{4}{17}\,e^{-\frac{1}{2}t}\,(4\cos(t)+\sin(t)) + \frac{4}{17}\,\mathrm{Heaviside}(t-\pi)\left(4\cos(t)-\sin(t)-e^{-\frac{1}{2}t+\frac{1}{2}\pi}\,(4\cos(t)+\sin(t))\right) \qquad (24)$$

```
> plot({G,Y},t=0..3*Pi);
```



Looking at Exercise 17

```
> assume(k>0);
> F:=(1/k)*(Heaviside(t-5)*(t-5)-Heaviside(t-5-k)*(t-5-k));
```

$$F := \frac{\text{Heaviside}(t-5)\ (t-5) - \text{Heaviside}(t-5-k\sim)\ (t-5-k\sim)}{k\sim} \quad \textbf{(25)}$$

```
> DE17:=diff(y(t),t$2)+4*y(t)=F;
```

$$DE17 := \frac{d^2}{dt^2}\ y(t) + 4\ y(t) = \frac{\text{Heaviside}(t-5)\ (t-5) - \text{Heaviside}(t-5-k\sim)\ (t-5-k\sim)}{k\sim} \quad \textbf{(26)}$$

```
> Eqn17:=laplace(DE17,t,s);
```

$$Eqn17 := s^2\ laplace(y(t), t, s) - D(y)(0) - s\ y(0) + 4\ laplace(y(t), t, s) = \frac{e^{-5s} - e^{-s(5+k\sim)}}{s^2\ k\sim} \quad \textbf{(27)}$$

```
> Eqn17A:=solve(Eqn17,laplace(y(t),t,s));
```

$$Eqn17A := \frac{s^3\ y(0)\ k\sim + D(y)(0)\ s^2\ k\sim + e^{-5s} - e^{-s(5+k\sim)}}{s^2\ k\sim\ (s^2+4)} \quad \textbf{(28)}$$

```
> Eqn17B:=subs({y(0)=0,D(y)(0)=0},Eqn17A);
```

$$Eqn17B := \frac{e^{-5s} - e^{-s(5+k\sim)}}{s^2 k\sim (s^2+4)} \quad \textbf{(29)}$$

```
> H:=1/(k*s^2*(s^2+4));
```

$$H := \frac{1}{k\sim s^2 (s^2+4)} \quad \textbf{(30)}$$

```
> H1:=convert(H,parfrac,s);
```

$$H1 := \frac{1}{4 s^2 k\sim} - \frac{1}{4 k\sim (s^2+4)} \quad \textbf{(31)}$$

```
> Yk:=invlaplace(H1*(exp(-5*s)-exp(-(5+k)*s)),s,t);
```

$$Yk := \frac{1}{8} \frac{1}{k\sim} \left((2t - \sin(2t-10) - 10) \operatorname{Heaviside}(t-5) - (-2k\sim + 2t + \sin(-2t+10 + 2k\sim) - 10) \operatorname{Heaviside}(t-5-k\sim) \right) \quad \textbf{(32)}$$

```
> Y1:=subs(k=2,Yk);
```

$$Y1 := \frac{1}{16} (2t - \sin(2t-10) - 10) \operatorname{Heaviside}(t-5) - \frac{1}{16} (-14 + 2t + \sin(-2t + 14)) \operatorname{Heaviside}(t-7) \quad \textbf{(33)}$$

```
> plot(Y1,t=0..30);
```




```
> Y2:=subs(k=5,Yk);
```

$$Y2 := \frac{1}{40}\left(2t - \sin(2t - 10) - 10\right)\text{Heaviside}(t - 5) - \frac{1}{40}\left(-20 + 2t + \sin(-2t + 20)\right)\text{Heaviside}(t - 10) \quad (34)$$

```
> plot({Y1,Y2},t=0..30);
```




```
> Y3:=subs(k=10,Yk);
```

$$Y3 := \frac{1}{80}\left(2t - \sin(2t - 10) - 10\right)\text{Heaviside}(t - 5) - \frac{1}{80}\left(-30 + 2t + \sin(-2t + 30)\right)\text{Heaviside}(t - 15) \quad (35)$$

```
> plot({Y1,Y2,Y3},t=0..30);
```




```
> ExpA:=(1/(8*k))*( (2*t-sin(2*(t-5))-10) - (-2*k+2*t-sin(2*(t-5-k))-10) );
```

$$ExpA := \frac{1}{8}\,\frac{-\sin(2\,t-10)+2\,k\sim-\sin(-2\,t+10+2\,k\sim)}{k\sim} \quad (36)$$

```
> simplify(ExpA);
```

$$-\frac{1}{8}\,\frac{\sin(2\,t-10)-2\,k\sim+\sin(-2\,t+10+2\,k\sim)}{k\sim} \quad (37)$$

```
> # To answer the remaining questions, it's easiest to use a formula for sin(A)+sin(B)
```